


# Timed symbolic transition system - Reference system model

SPTG automata models, called **timed symbolic transition systems**, will be introduced in the following using a small dummy one,  $\mathbb{G}$  sufficient to illustrate the main components:



Channels are  $C = \{ In, Out, Done \}$ , data variables are  $A = \{ x, sum \}$ , and clocks are  $K = \{ cl \}$ . Clock  $cl$  ranges over positive rationals, providing a dense time domain, and measures elapsed time. Variable  $x$  stores incoming values, while variable  $sum$  accumulates them (initialized to 0).

Channels are typed and declared as input or output:

- $In?x$  binds  $x$  to a value of the type of  $In$  (here, positive rationals),
- $Out!t$  emits a value  $t$  of the type of  $Out$ ,
- channels without data, like  $Done$ , have undefined type.

A **transition** is a tuple

$$(q, act, \phi, \mathbb{K}, \rho, q')$$

where:

- $q, q'$  are states,
- $act$  is an input/output action,
- $\phi$  is a guard,
- $\mathbb{K} \subseteq K$  is the set of clocks to reset,
- $\rho$  is an update  $x_1! := !t_1, \dots, x_n! := !t_n$  on data variables (where terms  $t_i$  may involve clocks).

When  $\rho = id$ , variables remain unchanged; only relevant updates are shown.

The system  $\mathbb{G}$  has three states  $q_0, q_1, q_2$  and four transitions:

- $\mathbf{tr}_1 = (q_0, In?x, 1 \leq x \leq 10, cl, sum! := !sum + x, q_1)$   
→ processes inputs;
- $\mathbf{tr}_2 = (q_1, Out!0, x \leq 5 \wedge cl = 42 - x, \emptyset, id, q_0)$   
→ emits 0 for small inputs;
- $\mathbf{tr}_3 = (q_1, Out!x, x > 5 \wedge cl = 42 - x, \emptyset, id, q_0)$   
→ emits the received value otherwise;
- $\mathbf{tr}_4 = (q_0, Done!, sum \geq 15, \emptyset, id, q_2)$   
→ may signal termination when the accumulated sum reaches 15.

## Encoding of the timed symbolic transition system in XLIA

The SPTG framework, being an extension of the DIVERSITY symbolic execution platform, uses its entry language to express models of timed symbolic transition systems. This language, called **XLIA (eXecutable Language for Interaction and Architecture)**, is designed for specifying the behavior of component-based systems.

In XLIA, components are referred to as machines. These machines are communicating, hierarchical, and heterogeneous, and their evaluation semantics can be customized to support different analysis or execution contexts.

The complete XLIA specification is available in the [XLIA documentation](#).

For timed symbolic transition systems, a single machine is typically used to represent the system under study, which interacts with its environment through well-defined communication interfaces.

The automaton  $\mathbb{G}$  is encoded in the **XLIA** input language of the **DIVERSITY** symbolic execution platform as follows:

```

timed system Example01_Dummy {

@composite:
  statemachine Example01_Dummy {
    @public:
      port input  In( urationa1 );
      port output Out( urationa1 );
      port output Done;
    @private:
      var urationa1      sum;
      var urationa1      x;
      var clock urationa1 cl;
    @region:

      state<start> q0 {
        @init{
          sum := 0;
        }
        transition tr1 --> q1 {
          input In( x );

```

```

        guard ( 1 <= x <= 10 );
        sum := sum+x;
        cl  := 0;
    }
    transition tr4 --> q2 {
        guard ( sum >= 15 );
        output Done;
    }
}
state q1 {
    transition tr2 --> q0 {
        guard ( x <= 5 && cl == 42-x );
        output Out( 0 );
    }
    transition tr3 --> q0 {
        guard ( x > 5 && cl == 42-x );
        output Out( x );
    }
}

state q2;

@com:
    connect< env >{
        input  In;
        output Out;
        output Done;
    }
}

```

## 1. General Structure

The **XLIA model** (entry language of the **DIVERSITY** symbolic execution platform) encodes the automaton by explicitly separating the **static part** (declarations of variables, clocks, and communication ports) from the **behavioral part** (states, transitions, and synchronization).

The **timed system** construct defines the **whole system**, while the nested **statemachine** block defines the actual automaton.

## 2. Static Part

The static part declares:

- **Ports** (inputs/outputs), which correspond to the external interactions of the automaton.
- **Variables** (data and clocks), which capture internal state and timing information.

From the XLIA model:

```

@public:
    port input  In( urational );

```

```
    port output Out( urational );
    port output Done;
    @private:
        var urational      sum;
        var urational      x;
        var clock urational cl;
```

Mapping to the automaton

XLIA element	Meaning	Automaton equivalent
port input In(urational)	<b>Input port</b> for receiving a rational value <i>x</i>	Transition input action <i>In?x</i>
port output Out(urational)	<b>Output port</b> for sending a rational value	Transition output action <i>Out!0</i> or <i>Out!x</i>
port output Done	<b>Output signal</b> without data	Transition output action <i>Done</i>
var urational sum	<b>Data variable</b> tracking accumulated input	Automaton variable (used in <b>guards/actions</b> )
var urational x	<b>Data variable</b> holding the latest input	<b>Variable bound</b> by input message
var clock urational cl	<b>Clock variable</b> (measures elapsed time since last reset)	Automaton clock for <b>timed constraints</b>

Thus, the static part declares the **interface** and **internal state space** of the automaton.

3. Behavioral Part

The behavioral description is under `@region`, where states and transitions are defined.

```
@region:
    state<start> q0 { ... }
    state q1 { ... }
    state q2;
```

States

- q0: Initial state (`<start>`), where sum is initialized.
- q1: Intermediate state reached after an input.
- q2: Final or terminal state when the condition `sum >= 15` holds.

Initialisation

```
@init {
    sum := 0;
}
```

Transitions from q0

```
transition tr1 --> q1 {
    input In( x );
    guard ( 1 <= x <= 10 );
    sum := sum+x;
    cl  := 0;
}
transition tr4 --> q2 {
    guard ( sum >= 15 );
    output Done;
}
```

Transition Mapping

Transition	Automaton equivalent	Description
tr1	$(q_0, In?x, 1 \leq x \leq 10, cl, sum! := sum + x, q_1)$	Receives input $x$ , adds to $sum$ , resets the clock $cl$ .
tr4	$(q_0, Done!, sum \geq 15, \emptyset, id, q_2)$	Produces Done output when accumulated $sum \geq 15$ .

Transitions form q1

```
transition tr2 --> q0 {
    guard ( x <= 5 && cl == 42-x );
    output Out( 0 );
}
transition tr3 --> q0 {
    guard ( x > 5 && cl == 42-x );
    output Out( x );
}
```

Transition	Automaton equivalent	Description
tr2	$(q_1, In?x, 1 \leq x \leq 10, cl, sum! := sum + x, q_1)$	Emits value 0, for small inputs.
tr3	$(q_1, Out!0, x \leq 5 \wedge cl = 42 - x, \emptyset, id, q_0)$	Emits received input $x$ , otherwise.

## 4. Communication Interface

The `@com` section defines how this state machine interacts with the environment:

```
@com:
  connect< env >{
    input  In;
    output Out;
    output Done;
  }
```

This binds the declared input/output ports to the environment.

## 5. Summary: Automaton -- XLIA Correspondence

Automaton Concept	XLIA Encoding
Locations $q_0, q_1, q_2$	<code>state q0, state q1, state q2</code>
Initial location	<code>state&lt;start&gt;</code>
Transitions	<code>transition trX --&gt; qY { ... }</code>
Guards	<code>guard(...)</code> expressions
Clocks	Declared with <code>var clock</code>
Clock resets	Assignments like <code>c1 := 0</code>
Input actions	<code>input In(x)</code>
Output actions	<code>output Out(...)</code> or <code>output Done</code>
Variable updates	Direct assignments inside transition body
Terminal state	<code>state q2;</code> (no outgoing transitions)

### More on XLIA subset to encode timed symbolic transition system

```
// =====
// Prologue - Header
// =====
@xlia< system , 1.0 >:

// =====
// System Definition
// =====
timed system S {

  // =====
```

```

// Composite Part: State Machine Definition
// =====
@composite:
statemachine SM {
    @public:

        // -----
        // Declaration of Ports
        // -----
        port input  In;
        port output Done;
        port input  In1( urationa1 );
        port input  In2( integer );
        port output Out( urationa1 );

        // Declaration of N-ary Ports
        port input  In3( bool, integer, rationa1 );
        port output Out2( integer, bool );

        // -----
        // Declaration of Constants
        // -----
        const integer N = 42;

    @private:

        // -----
        // Declaration of Variables
        // -----
        var urationa1 sum;
        var urationa1 x;
        var urationa1 y;
        var integer   z;
        var bool      flag;
        var integer   fee;

        // -----
        // Declaration of Clocks
        // -----
        var clock urationa1 cl;
        var clock urationa1 cl2;

// =====
// Behavioral Description: States and Transitions
// =====
@region:

    // -----
    // Initial State
    // -----
    state<start> q0 {
        @init {
            sum := 0;
            flag := false;

```

```

        guard( fee > 0 );
    }

    transition tr1 --> q1 {
        input In1( x );
        guard ( 1 <= x <= 10 );
        sum := sum + x;
        y   := sum;
        cl  := 0;
    }

    transition tr2 --> q1 {
        input In( x );
        guard ( 10 < x && x < N );
        { |, |
            sum := sum + x;
            y   := sum; // y receives the pre-increment sum value
        }
        cl2 := 0;
    }
}

// -----
// Secondary State
// -----
state q1 {
    transition tr3 --> q0 {
        guard( x <= 10 && cl == N - x );
        output Out( sum - 1 );
    }

    transition tr4 --> q0 {
        guard( x > 10 );
        guard( cl <= 5 );
        output Out2( fee, flag );
        flag := true;
        cl2 := 0;
    }

    transition tr5 --> q2 {
        guard( sum >= 15 && cl2 <= 1 );
        output Done;
        cl2 := 0;
    }
}

// -----
// Terminal State
// -----
state q2;

// =====
// Communication Part: Port Connections
// =====

```



```
@com:
connect< env > {
  input  In;
  input  In1;
  input  In2;
  input  In3;
  output Done;
  output Out;
  output Out2;
}
}
```